////////////////////////////////////////////////////////////////////////////////
//
//  ADOBE SYSTEMS INCORPORATED
//  Copyright 2004-2007 Adobe Systems Incorporated
//  All Rights Reserved.
//
//  NOTICE: Adobe permits you to use, modify, and distribute this file
//  in accordance with the terms of the license agreement accompanying it.
//
////////////////////////////////////////////////////////////////////////////////

package com.lowke.animator.effect.tween.easing.adobe
{
    /**
     *  The Circular class defines three easing functions to implement 
     *  circular motion with Flex effect classes. 
     *
     *  For more information, see http://www.robertpenner.com/profmx.
     */  
    public class Circular
    {
        //--------------------------------------------------------------------------
        //
        //  Class methods
        //
        //--------------------------------------------------------------------------
        
        /**
         *  The <code>easeIn()</code> method starts motion slowly, 
         *  and then accelerates motion as it executes. 
         *
         *  @param t Specifies time.
         *
         *  @param b Specifies the initial position of a component.
         *
         *  @param c Specifies the total change in position of the component.
         *
         *  @param d Specifies the duration of the effect, in milliseconds.
         *
         *  @return Number corresponding to the position of the component.
         */  
        public static function easeIn(t:Number, b:Number,
                                      c:Number, d:Number):Number
        {
            return -c * (Math.sqrt(1 - (t /= d) * t) - 1) + b;
        }
        
        /**
         *  The <code>easeOut()</code> method starts motion fast, 
         *  and then decelerates motion as it executes. 
         *
         *  @param t Specifies time.
         *
         *  @param b Specifies the initial position of a component.
         *
         *  @param c Specifies the total change in position of the component.
         *
         *  @param d Specifies the duration of the effect, in milliseconds.
         *
         *  @return Number corresponding to the position of the component.
         */  
        public static function easeOut(t:Number, b:Number,
                                       c:Number, d:Number):Number
        {
            return c * Math.sqrt(1 - (t = t/d - 1) * t) + b;
        }
        
        /**
         *  The <code>easeInOut()</code> method combines the motion
         *  of the <code>easeIn()</code> and <code>easeOut()</code> methods
         *  to start the motion slowly, accelerate motion, then decelerate. 
         *
         *  @param t Specifies time.
         *
         *  @param b Specifies the initial position of a component.
         *
         *  @param c Specifies the total change in position of the component.
         *
         *  @param d Specifies the duration of the effect, in milliseconds.
         *
         *  @return Number corresponding to the position of the component.
         */  
        public static function easeInOut(t:Number, b:Number,
                                         c:Number, d:Number):Number
        {
            if ((t /= d / 2) < 1)
                return -c / 2 * (Math.sqrt(1 - t * t) - 1) + b;
            
            return c / 2 * (Math.sqrt(1 - (t -= 2) * t) + 1) + b;
        }
    }
}



/* ORIGINAL VERSION
// Easing functions 
//  t = time, b = begin, c = range, d = duration

public class Circular {

// accelerating from zero velocity
public static function easeIn (t:Number, b:Number, c:Number, d:Number):Number {
return -c * (Math.sqrt(1 - (t/=d)*t) - 1) + b;
}
// decelerating to zero velocity
public static function easeOut (t:Number, b:Number, c:Number, d:Number):Number {
return c * Math.sqrt(1 - (t=t/d-1)*t) + b;
}
// acceleration until halfway, then deceleration 
public static function easeInOut (t:Number, b:Number, c:Number, d:Number):Number {
if ((t/=d/2) < 1) return -c/2 * (Math.sqrt(1 - t*t) - 1) + b;
return c/2 * (Math.sqrt(1 - (t-=2)*t) + 1) + b;
}
}
*/
